

Financial Econometrics Review Session Notes 10

Nina Boyarchenko

March 12, 2010

Contents

1	Factor Model	2
2	Conditional Portfolio Weights	3
3	DCC model	4

In this review session, we will be working with the 6 FF portfolio returns (returns.txt). We will also need two program files: **opt_port.prg**, which calculates the optimal portfolio weights and returns, and **dcc.prg**, which estimates the dcc model.

We begin by loading the data set and converting the raw returns on the portfolios into excess returns. Notice that the column **mkt_rf** contains the excess market return and the column **rf** the annualized risk-free rate.

```
series mse=ms-rf/52
series mbe=mb-rf/52
```

We will construct portfolio returns for a strategy that trades in the Mid-Value, Small Stock portfolio (**MS**) and Mid-Value, Large Stock portfolio (**MB**).

1 Factor Model

Consider estimating a factor model for MS and MB. Recall that this is given by:

$$r_{it}^e = \beta_{0i} + \beta_{1i}r_{Mt}^e + \epsilon_{it}, \quad i = \{MS, MB\},$$

where the superscript e denotes excess returns. In EViews, we use:

```
vector(2) beta0
vector(2) beta1
vector(2) nu
```

```
equation lsms.ls mse c mkt_rf
beta0(1)=c(1)
beta1(1)=c(2)
series residms=resid
nu(1)=@var(residms)
```

```
equation lsmb.ls mbe c mkt_rf
beta0(2)=c(1)
beta1(2)=c(2)
series residmb=resid
nu(2)=@var(residmb)
```

Consider now estimating a GARCH model for the excess market return. Recall that, for GARCH estimation, we use the **arch** function:


```
equation garch11.arch(1,1) mkt_rf
garch11.makegarch garch11var 'save the conditional GARCH variance
series garch11vol=@sqrt(garch11var) 'compute the corresponding vol
series garch11err=mkt_rf/garch11vol 'compute the standardized residuals
```

Then, to construct the conditional correlation between MS and MB, we would use:

```
series corr=beta1(1)*beta1(2)*garch11var/@sqrt((beta1(1)^2*garch11var+nu(1))
*(beta1(2)^2*garch11var+nu(2)))
series corr_fact=corr
```

The conditional correlation is plotted in Fig. 1. Notice that, to compute the variance of

Figure 1: Conditional Correlation



a portfolio which has α invested in the MS portfolio and $(1 - \alpha)$, we use:

$$var(\alpha r_{MS}^e + (1 - \alpha)r_{MB}^e) = \alpha^2(\beta_{1,MS}^2 h_t + \nu_{MS}) + (1 - \alpha)^2(\beta_{1,MB}^2 h_t + \nu_{MB}) + 2\sqrt{\alpha(1 - \alpha)}h_t\rho_t$$

2 Conditional Portfolio Weights

Consider now calculating optimal portfolio weights for the strategy that trades in MS and MB. The program **opt_port.prg** calculates the returns for the unconditional weights and weights given by the correlation time series. In particular, the program file contains a subroutine called **portfolio**, which takes the desired level of return (μ_0) and the start point of the calculation as an argument. To call this in EViews, we would use:

```
series hx=beta1(1)^2*garch11var+nu(1)
series hy=beta1(2)^2*garch11var+nu(2)
series x=mse
series y=mbe
```

```
include opt_port
call portfolio(0.05,1)
series ret_fact=ret
```

Notice that we need to provide the conditional variance time series for the routine.

3 DCC model

Now, consider estimating a DCC model for the excess returns. For this, we will use the `dcc` program. In particular, we use:

```
include dcc
series corr_dcc=corr
```

Then, to calculate the optimal weights and the corresponding returns, we call once again the `portfolio` subroutine:

```
call portfolio(0.05, 2)
series ret_dcc=ret
```

The returns to the three trading strategies are plotted in Fig. 2.

Figure 2: Portfolio returns

